
1. [bookmark: _Toc122434485][bookmark: _Toc66458179][bookmark: _Toc85580837]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.6.1.3.1 which is part of the NR5GC test suite.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com
	Shaun Harry
	shaun.harry@keysight.com
0. [bookmark: _Toc51622099][bookmark: _Toc55479332][bookmark: _Toc60751530][bookmark: _Toc60757075][bookmark: _Toc85580838]Verification Test Summary 
Test Case: 	8.1.6.1.3.1
ATS Version:	iWD-TTCN3-B2020-09_D21wk37
[bookmark: _Hlk37919671]System Simulator used:	Anritsu Protocol Conformance Test System ME7834NR
UE used:	Qualcomm SM 8450 + SDX65
Verification Status:	PASS


1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc85580839]Corrections required
1.1 [bookmark: _Toc76738641][bookmark: _Toc85580840]Change 1
	Function name
	f_TC_8_1_6_1_3_1_NR5GC()

	Reason for change
	The function f_TC_8_1_6_1_3_Common() is used by both TC 8.1.6.1.3.1. It needs to be informed whether the measurements to be made is either intra- or inter-frequency. 

	Summary of change
	Set to “true” the new input parameter added to f_TC_8_1_6_1_3_Common(). See Change 2.2.

	TTCN module
	RRC_MDT_IntraNR_RLF_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	function f_TC_8_1_6_1_3_2_NR5GC() runs on NR5GC_PTC
  { /* Radio Link Failure / Reporting of Inter-frequency measurements */
  
    f_NR5GC_Init(NR_4);
    f_TC_8_1_6_1_3_Common(nr_Cell3);
      
  }// End of f_TC_8_1_6_1_3_2_NR5GC()



After Change:
	function f_TC_8_1_6_1_3_2_NR5GC() runs on NR5GC_PTC
  { /* Radio Link Failure / Reporting of Inter-frequency measurements */
  
    f_NR5GC_Init(NR_4);
    f_TC_8_1_6_1_3_Common(nr_Cell3, -, true);
      
  }// End of f_TC_8_1_6_1_3_2_NR5GC()




1.2 [bookmark: _Toc76738642][bookmark: _Toc85580841]Change 2
	Function name
	f_TC_8_1_6_1_3_Common()

	Reason for change
	This common function is used by both TC 8.1.6.1.3.1 (intra freq) and TC 8.1.6.1.3.2 (inter frequency). It needs to know whether to target measurement frequency is intra or inter to create the correct Measurement-related IEs in RrcReconfiguration message.

For Inter-Frequency measurements, Source and Target ARFCNs are needed. Measurement Gap is also needed.

Typo error. Target Cell for Reestablishment is the Neighbor cell.

c_RNTI_r16 of nr_RLF_Report_r16 must be that of the Cell where RLF happened – in this case nr_Cell1. Original TTCN has default to check c_RNTI_r16 using the RNTI for current serving cell.

TS 38.523-1 states that UEInformationResponse containing no “logMeasReport” is to be transmitted via SRB1 as below:

1>	if the logMeasReport is included in the UEInformationResponse:
2>	submit the UEInformationResponse message to lower layers for transmission via SRB2;
2>	discard the logged measurement entries included in the logMeasInfoList from VarLogMeasReport upon successful delivery of the UEInformationResponse message confirmed by lower layers;
1>	else:
2>	submit the UEInformationResponse message to lower layers for transmission via SRB1.

	Summary of change
	Added an input parameter of Boolean type p_InterFrequencyTarget with default value false

Assignment expressions for variables used to generate the MeasConfig IE in RRCReconfiguration message were modified depending whether the p_InterFrequencyTarget is true or false

The 2nd Input parameter to f_NR5GC_RRCReestablishment_SON_MDT() was changed from nr_Cell1 to p_NeighbourCell

When checking UEInformationResponse at step12, instead of using ‘-‘, used C_RNTI value of nr_Cell1 as input value to parameter p_RNTI_Value of template cr_UEInformationResponse_r16_RLF

Changed TTCN to expect UEInformationResponse on SRB1 instead of SRB2

	TTCN module
	RRC_MDT_IntraNR_RLF_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	function f_TC_8_1_6_1_3_Common(NR_CellId_Type p_NeighbourCell, boolean p_CheckLocationInfo := false) runs on NR5GC_PTC
  {
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var NR_AbsoluteCellPower_Type v_CellPower_A3_Entry_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_A3_Entry_FR2;
    
    var RSRP_Range v_RsrpThres;
    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasIdToAddModList v_MeasIdConfigList;
    var template (omit) MeasConfig v_MeasConfig;
    var PhysCellId v_PhysCellIdServing;
    var PhysCellId v_PhysCellIdNeighbour;
    var ARFCN_ValueNR v_ARFCN_Serving;
    var ARFCN_ValueNR v_ARFCN_Neighbour;
    var template (present) UL_DCCH_Message v_UEInformationResponse;
    var DeltaValues_Type v_Deltas := f_NR_MobileInfo_GetPrimaryDeltaValue();  //@sic R5-214550 sic@

    var CellIdentity v_CellIdentity_Neighbour;
      
    //Table 8.1.6.1.3.1.3.2-1/2
    v_CellPower_A3_Entry_FR1 := -82;
    v_CellPower_A3_Entry_FR2 := -73;
    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2),
                             cs_NR_CellPower(p_NeighbourCell, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2)};
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_NeighbourCell, v_CellPower_A3_Entry_FR1, v_CellPower_A3_Entry_FR2)};
    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)};
    
    f_NR_CellInfo_InitMaxReferencePower(p_NeighbourCell, v_CellPower_A3_Entry_FR1, v_CellPower_A3_Entry_FR2);
       
    //Create and configure cells
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(p_NeighbourCell);
    
    //Preamble
    //The UE is in state 3N-A as defined in TS 38.508-1 [4], subclause 4.4A on NR Cell 1.
    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    if(p_CheckLocationInfo)
    {/*The UE's positioning engine (e.g. standalone GNSS receiver) should be provided with any necessary stimulus to allow it to provide the position.
      *This shall be done by use of the test function Update UE Location Information defined in TS 38.509 [6], if supported by the UE according to pc_UpdateUE_LocationInformation */
        f_NR_ProvideUE_LocationInformation(nr_Cell1, RRC_CONNECTED);
    }


    f_NR_TestBody_Set(true);
    
    f_NR_SetCellPowerList(v_CellPowerList_AtT0);
    
    //Table 8.1.6.1.3.1.3.3-2|3|4
    v_RsrpThres := 2;  //@sic R5-216289 sic@
    if (p_NeighbourCell == nr_Cell3)
    {
        v_RsrpThres := v_RsrpThres + v_Deltas.DeltaNRf2 - v_Deltas.DeltaNRf1 //@sic R5-214550 sic@
    }


          
    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1)};
    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_RsrpThres, infinity_, -, -, cs_NR_MeasReportQuantity_Rsrp)};
    v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
    
    v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);

    //@siclog "Steps 1-2" siclog@
    // The SS transmits an RRCReconfiguration message including MeasConfig to setup NR measurement and reporting for intra-frequency event A3 (measId 1).
    //The UE transmits an RRCReconfigurationComplete message.
    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);
    
    //@siclog "Step 3" siclog@
    //The SS changes NR Cell 2 parameters according to the row "T1" in Table 8.1.6.1.3.1.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);
    
    //@siclog "Step 4" siclog@
    //The UE transmit a MeasurementReport message to report event A3 (measId 1) with the measured RSRP value for NR Cell 2.
    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(p_NeighbourCell);
    f_NR_ReceiveMeasurementReports(nr_Cell1,
                                  v_PhysCellIdServing,
                                  tsc_NR_MeasId1,
                                  -,
                                  -,
                                  cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));
    //@siclog "Step 5" siclog@
    //The SS changes NR Cell 1 parameter according to the row "T2" in Table 8.1.6.1.3.1.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);
      
    //@siclog "Steps 6-10" siclog@
    //The UE transmits an RRCReestablishmentRequest message on neighblur cell.
    //The SS transmits an RRCReestablishment message.
    //The UE transmits an RRCReestablishmentComplete message.
    //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB.
    //The UE transmits an RRCReconfigurationtComplete message.
    //The SS transmits an RRCReestablishment message.
    //The UE transmits an RRCReestablishmentComplete message.
    f_NR5GC_RRCReestablishment_SON_MDT(nr_Cell1, nr_Cell1, rlf_InfoAvailable);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test " & testcasename() & " Step 9: RRCReestablishmentComplete message with rlf-InfoAvailable included");
       
    //@siclog "Step 11" siclog@
    //The SS transmits a UEInformationRequest message.
    SRB.send(cas_NR_SRB_RrcPdu_REQ(p_NeighbourCell,
                                  tsc_NR_RbId_SRB2,
                                  -, //cs_TimingInfo_Now
                                  cs_NR_UEInformationRequest_RLF)); // condition RLF
      
    //@siclog "Step 12" siclog@
    //Check: Does the UE transmit a UEInformationResponse message on NR Cell 2?
    v_ARFCN_Serving := f_NR_CellInfo_GetFrequencySSB(nr_Cell1);
    v_ARFCN_Neighbour := f_NR_CellInfo_GetFrequencySSB(p_NeighbourCell);
    v_CellIdentity_Neighbour := f_NR_CellInfo_GetCellIdentity(p_NeighbourCell);
    v_UEInformationResponse := cr_UEInformationResponse_r16_RLF(v_PhysCellIdServing, v_PhysCellIdNeighbour, v_ARFCN_Serving, v_ARFCN_Neighbour, v_CellIdentity_Neighbour);
    if(p_CheckLocationInfo)
    {
        v_UEInformationResponse.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.rlf_Report_r16.nr_RLF_Report_r16.locationInfo_r16 := ?;
    }
    SRB.receive(car_NR_SRB_RrcPdu_IND(p_NeighbourCell, tsc_NR_RbId_SRB2, v_UEInformationResponse));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test " & testcasename() & " Step 12");
      
    //@siclog "Step 13" siclog@
    //Void @sic R5-216289 sic@
      
    f_NR_TestBody_Set(false);
    // Finish
    f_NR_Postamble(p_NeighbourCell, STATE_CONNECTED_3A); // @sic R5-216289 sic@
  } // end of f_TC_8_1_6_1_3_Common



After Change:
	function f_TC_8_1_6_1_3_Common(NR_CellId_Type p_NeighbourCell, boolean p_CheckLocationInfo := false, boolean p_InterFrequencyTarget := false) runs on NR5GC_PTC
  {
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var NR_AbsoluteCellPower_Type v_CellPower_A3_Entry_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_A3_Entry_FR2;
    
    var RSRP_Range v_RsrpThres;
    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasIdToAddModList v_MeasIdConfigList;
    var template (omit) MeasConfig v_MeasConfig;
    var PhysCellId v_PhysCellIdServing;
    var PhysCellId v_PhysCellIdNeighbour;
    var ARFCN_ValueNR v_ARFCN_Serving;
    var ARFCN_ValueNR v_ARFCN_Neighbour;
    var template (present) UL_DCCH_Message v_UEInformationResponse;
    var DeltaValues_Type v_Deltas := f_NR_MobileInfo_GetPrimaryDeltaValue();  //@sic R5-214550 sic@
    var MeasGapConfig v_MeasGapConfig; //default  config
    var CellIdentity v_CellIdentity_Neighbour;
      
    //Table 8.1.6.1.3.1.3.2-1/2
    v_CellPower_A3_Entry_FR1 := -82;
    v_CellPower_A3_Entry_FR2 := -73;
    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2),
                             cs_NR_CellPower(p_NeighbourCell, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2)};
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_NeighbourCell, v_CellPower_A3_Entry_FR1, v_CellPower_A3_Entry_FR2)};
    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)};
    
    f_NR_CellInfo_InitMaxReferencePower(p_NeighbourCell, v_CellPower_A3_Entry_FR1, v_CellPower_A3_Entry_FR2);
       
    //Create and configure cells
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(p_NeighbourCell);
    
    //Preamble
    //The UE is in state 3N-A as defined in TS 38.508-1 [4], subclause 4.4A on NR Cell 1.
    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    if(p_CheckLocationInfo)
    {/*The UE's positioning engine (e.g. standalone GNSS receiver) should be provided with any necessary stimulus to allow it to provide the position.
      *This shall be done by use of the test function Update UE Location Information defined in TS 38.509 [6], if supported by the UE according to pc_UpdateUE_LocationInformation */
        f_NR_ProvideUE_LocationInformation(nr_Cell1, RRC_CONNECTED);
    }


    f_NR_TestBody_Set(true);
    
    f_NR_SetCellPowerList(v_CellPowerList_AtT0);
    
    //Table 8.1.6.1.3.1.3.3-2|3|4
    v_RsrpThres := 2;  //@sic R5-216289 sic@
    if (p_NeighbourCell == nr_Cell3)
    {
        v_RsrpThres := v_RsrpThres + v_Deltas.DeltaNRf2 - v_Deltas.DeltaNRf1 //@sic R5-214550 sic@
    }


    if(not p_InterFrequencyTarget){
        v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1)};     
        v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_RsrpThres, r1, -, -, cs_NR_MeasReportQuantity_Rsrp)};
        /*if (p_CheckLocationInfo){
              v_ReportConfigList[0].reportConfig.reportConfigNR.reportType.eventTriggered.includeCommonLocationInfo_r16 := true_;
            }*/
        v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj1_conf1};
        
        v_MeasConfig :=f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList);
    
     }
     else{
        v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1), cs_MeasObjectId2(p_NeighbourCell)};
        v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_RsrpThres, r1, -, -, cs_NR_MeasReportQuantity_Rsrp)};
        /*if (p_CheckLocationInfo){
          v_ReportConfigList[0].reportConfig.reportConfigNR.reportType.eventTriggered.includeCommonLocationInfo_r16 := true_;
        }*/
        v_MeasIdConfigList := {cs_NR_MeasId_Config_id1_obj2_conf1};
        //v_MeasGapConfig := cs_NR_MeasGapConfig_FR1(cs_38508_GapConfig);

        v_MeasGapConfig := cs_38508_MeasGapConfig_UE(f_NR_GetGapConfig_InterFreqOdd(nr_Cell1, p_NeighbourCell));
        v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, v_MeasIdConfigList, v_MeasGapConfig);
    }
    //@siclog "Steps 1-2" siclog@
    // The SS transmits an RRCReconfiguration message including MeasConfig to setup NR measurement and reporting for intra-frequency event A3 (measId 1).
    //The UE transmits an RRCReconfigurationComplete message.
    f_NR_SendRRCReconfiguration_MeasNR(nr_Cell1, v_MeasConfig);
    
    //@siclog "Step 3" siclog@
    //The SS changes NR Cell 2 parameters according to the row "T1" in Table 8.1.6.1.3.1.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);
    
    //@siclog "Step 4" siclog@
    //The UE transmit a MeasurementReport message to report event A3 (measId 1) with the measured RSRP value for NR Cell 2.
    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(p_NeighbourCell);
    f_NR_ReceiveMeasurementReports(nr_Cell1,
                                  v_PhysCellIdServing,
                                  tsc_NR_MeasId1,
                                  -,
                                  -,
                                  cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));
    //@siclog "Step 5" siclog@
    //The SS changes NR Cell 1 parameter according to the row "T2" in Table 8.1.6.1.3.1.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);
      
    //@siclog "Steps 6-10" siclog@
    //The UE transmits an RRCReestablishmentRequest message on neighblur cell.
    //The SS transmits an RRCReestablishment message.
    //The UE transmits an RRCReestablishmentComplete message.
    //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB.
    //The UE transmits an RRCReconfigurationtComplete message.
    //The SS transmits an RRCReestablishment message.
    //The UE transmits an RRCReestablishmentComplete message.
    f_NR5GC_RRCReestablishment_SON_MDT(nr_Cell1, p_NeighbourCell, rlf_InfoAvailable);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test " & testcasename() & " Step 9: RRCReestablishmentComplete message with rlf-InfoAvailable included");
       
    //@siclog "Step 11" siclog@
    //The SS transmits a UEInformationRequest message.
    SRB.send(cas_NR_SRB_RrcPdu_REQ(p_NeighbourCell,
                                  tsc_NR_RbId_SRB2,
                                  -, //cs_TimingInfo_Now
                                  cs_NR_UEInformationRequest_RLF)); // condition RLF
      
    //@siclog "Step 12" siclog@
    //Check: Does the UE transmit a UEInformationResponse message on NR Cell 2?
    v_ARFCN_Serving := f_NR_CellInfo_GetFrequencySSB(nr_Cell1);
    v_ARFCN_Neighbour := f_NR_CellInfo_GetFrequencySSB(p_NeighbourCell);
    v_CellIdentity_Neighbour := f_NR_CellInfo_GetCellIdentity(p_NeighbourCell);
    v_UEInformationResponse := cr_UEInformationResponse_r16_RLF(v_PhysCellIdServing, v_PhysCellIdNeighbour, v_ARFCN_Serving, v_ARFCN_Neighbour, v_CellIdentity_Neighbour, -, f_NR_CellInfo_GetRNTI(nr_Cell1));
    if(p_CheckLocationInfo)
    {
        v_UEInformationResponse.message_.messageClassExtension.c2.ueInformationResponse_r16.criticalExtensions.ueInformationResponse_r16.rlf_Report_r16.nr_RLF_Report_r16.locationInfo_r16 := ?;
    }
    SRB.receive(car_NR_SRB_RrcPdu_IND(p_NeighbourCell, tsc_NR_RbId_SRB1, v_UEInformationResponse));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Test " & testcasename() & " Step 12");
      
    //@siclog "Step 13" siclog@
    //Void @sic R5-216289 sic@
      
    f_NR_TestBody_Set(false);
    // Finish
    f_NR_Postamble(p_NeighbourCell, STATE_CONNECTED_3A); // @sic R5-216289 sic@
  } // end of f_TC_8_1_6_1_3_Common

	







1.3 [bookmark: _Toc85580842]Change 3
	Function name
	template RLF_Report_r16 cr_NR_RLF_Report_r16

	Reason for change
	Original prose/TTCN expects only ARFCN/PCI info for the failedPCell ie. However, spec allows for either CGIInfo/TAC info or ARFCN/PCI of the failedPcell to be reported during RLF as stated below from TS38.331:

1> else if the failure is detected due to radio link failure as described in 5.3.10.3, set the fields in VarRLF-report as follows:
2> set the connectionFailureType to rlf;
2> set the rlf-Cause to the trigger for detecting radio link failure in accordance with clause 5.3.10.4;
2> set the nrFailedPCellId in failedPCellId to the global cell identity and the tracking area code, if available, and otherwise to the physical cell identity and carrier frequency of the PCell where radio link failure is detected;
# R5-216823 is raised during RAN5#93-e meeting

	Summary of change
	Modified the expected UEInformationResponse to contain CGIInfo of failed cell. See Change 2.3 for additional details


	TTCN module
	NR5GC_MDT_Specific_Templates.ttcn

	MCC160 Comment
	



Before Change:
	template(present) RLF_Report_r16 cr_NR_RLF_Report_r16 (PhysCellId p_PhysCellId_Old,
                                                  PhysCellId p_PhysCellId_New,
                                                  ARFCN_ValueNR p_ARFCN_ValueNR_Old,
                                                  ARFCN_ValueNR p_ARFCN_ValueNR_New,
                                                  CellIdentity p_CellIdentity,
                                                  template (present) SSB_Index p_SSB_Index := 1,
                                                  RNTI_Value p_RNTI_Value := tsc_NR_RRC_TI_Def) :=
    {
        nr_RLF_Report_r16 :=
       {
           measResultLastServCell_r16 := cr_MeasResultRLFNR_r16_1SSB(p_SSB_Index),
           measResultNeighCells_r16 := cr_NR_MeasResultNeighCells_r16_NRList ({cr_NR_MeasResult2NR_r16 (p_PhysCellId_New, p_ARFCN_ValueNR_New)}),
           c_RNTI_r16 := p_RNTI_Value,
           previousPCellId_r16 := omit,
           failedPCellId_r16 :=
           {
               nrFailedPCellId_r16 :=
               {
                   pci_arfcn_r16 := cr_NR_RLF_PCI_ARFCN_r16(p_PhysCellId_Old, p_ARFCN_ValueNR_Old)
               }
           },
           reconnectCellId_r16 := omit,
           timeUntilReconnection_16 := omit,
           reestablishmentCellId_r16 := cr_NR_RLF_ReestablishmentCellId(p_CellIdentity),
           timeConnFailure_r16 := omit,
           timeSinceFailure_r16 := ?,
           connectionFailureType_r16 := rlf,
           rlf_Cause_r16 := t310_Expiry,
           locationInfo_r16 := *,
           noSuitableCellFound_r16 := omit,
           ra_InformationCommon_r16 := omit
        }
    };



After Change:
	template(present) RLF_Report_r16 cr_NR_RLF_Report_r16 (PhysCellId p_PhysCellId_Old,
                                                  PhysCellId p_PhysCellId_New,
                                                  ARFCN_ValueNR p_ARFCN_ValueNR_Old,
                                                  ARFCN_ValueNR p_ARFCN_ValueNR_New,
                                                  CellIdentity p_CellIdentity,
                                                  template (present) SSB_Index p_SSB_Index := 1,
                                                  RNTI_Value p_RNTI_Value := tsc_NR_RRC_TI_Def) :=
    {
        nr_RLF_Report_r16 :=
       {
           measResultLastServCell_r16 := cr_MeasResultRLFNR_r16_1SSB(p_SSB_Index),
           measResultNeighCells_r16 := cr_NR_MeasResultNeighCells_r16_NRList ({cr_NR_MeasResult2NR_r16 (p_ARFCN_ValueNR_New,p_PhysCellId_New)}),
           c_RNTI_r16 := p_RNTI_Value,
           previousPCellId_r16 := omit,
           failedPCellId_r16 :=
           {
               nrFailedPCellId_r16 :=?
               /*{
                   pci_arfcn_r16 := cr_NR_RLF_PCI_ARFCN_r16(p_PhysCellId_Old, p_ARFCN_ValueNR_Old)
               }*/ 
           },
           reconnectCellId_r16 := omit,
           timeUntilReconnection_16 := omit,
           reestablishmentCellId_r16 := cr_NR_RLF_ReestablishmentCellId(p_CellIdentity),
           timeConnFailure_r16 := omit,
           timeSinceFailure_r16 := ?,
           connectionFailureType_r16 := rlf,
           rlf_Cause_r16 := t310_Expiry,
           locationInfo_r16 := *,
           noSuitableCellFound_r16 := omit,
           ra_InformationCommon_r16 := omit
        }
    };
    

	





1.4 [bookmark: _Toc85580843]Change 4
	Function name
	template () MeasResult2NR_r16 cr_NR_MeasResult2NR_r16 ()

	Reason for change
	The template cr_NR_MeasResultNR used to populate the measResultList_r16 IE in this template requires that the rsIndexResults be OMITTED. See NR_Measurement_Templates.ttcn. However, according to TS 38.523-1, all optional measurement IEs may be reported by the UE during RLF scenarios:

1> for each of the configured measObjectNR in which measurements are available:
2> if the SS/PBCH block-based measurement quantities are available:
3> set the measResultListNR in measResultNeighCells to include all the available measurement quantities of the best measured cells, other than the source PCell(in case HO failure) or PCell (in case RLF), ordered such that the cell with highest SS/PBCH block RSRP is listed first if SS/PBCH block RSRP measurement results are available, otherwise the cell with highest SS/PBCH block RSRQ is listed first if SS/PBCH block RSRQ measurement results are available, otherwise the cell with highest SS/PBCH block SINR is listed first, based on the available SS/PBCH block based measurements collected up to the moment the UE detected failure;
4> for each neighbour cell included, include the optional fields that are available;

# R5-216823 is raised during RAN5#93-e meeting

	Summary of change
	Used a new template to generate expected MeasResultList. See Change 2.6.

	TTCN module
	NR5GC_MDT_Specific_Templates.ttcn

	MCC160 Comment
	



Before Change:
	template (present) MeasResult2NR_r16 cr_NR_MeasResult2NR_r16 (ARFCN_ValueNR p_ARFCN_ValueNR, PhysCellId p_PhysCellId) :=
    {
       ssbFrequency_r16 :=  p_ARFCN_ValueNR,
       refFreqCSI_RS_r16 := omit,
       measResultList_r16 := {cr_NR_MeasResultNR_CellResults(p_PhysCellId)}
    } ;



After Change:
	template (present) MeasResult2NR_r16 cr_NR_MeasResult2NR_r16 (ARFCN_ValueNR p_ARFCN_ValueNR, PhysCellId p_PhysCellId) :=
    {
       ssbFrequency_r16 :=  p_ARFCN_ValueNR,
       refFreqCSI_RS_r16 := omit,
       measResultList_r16 := {cr_NR_MeasResultNR_CellResults_WithRsIndex(p_PhysCellId)}
    } ;

	





1.5 [bookmark: _Toc85580844]Change 5
	Function name
	template () MeasResultNR cr_NR_MeasResultNR_CellResults_WithRsIndex

	Reason for change
	Same as 2.5

	Summary of change
	Created new template based on the old cr_NR_MeasResultNR_CellResults template, but with the IE rsIndexResults set to ANYOROMIT

	TTCN module
	NR5GC_MDT_Specific_Templates.ttcn

	MCC160 Comment
	



Before Change:
	None. New template added.



After Change:
	template (present) MeasResultNR cr_NR_MeasResultNR_CellResults_WithRsIndex(PhysCellId p_PhysCellId,
                                                                 template  (present)MeasQuantityResults p_MeasQuantityResults := f_NR_Get_MeasQuantityResults()
  ):=    //@sic R5s190416 sic@
  { /* @status    APPROVED (ENDC, NR5GC, NR5GC_IRAT) */
    physCellId := p_PhysCellId,
    measResult := {
      cellResults := {
        resultsSSB_Cell := p_MeasQuantityResults,
        resultsCSI_RS_Cell := omit
      },
      rsIndexResults := *    //@sic R5-204438 sic@
    },
    cgi_Info := omit
  };

	





1.6 [bookmark: _Toc85580845]Change 6
	Function name
	f_GetTestcaseAttrib_EarlyContentionResolution

	Reason for change
	TS 38.523-3 cl. 7.3.5.1 specifies that for specific test purposes, it is necessary to send DL CCCH/DCCH message separately from MSG4 as stated below:


There are cases however where it is necessary to send the DL CCCH or DL DCCH message separately from RA Msg4, this is implemented in TTCN using the test case attribute EarlyContentionResolution:
-	RRC connection establishment: when RRCSetup message is part of the test purpose,
-	RRC connection reject: when RRCReject message is part of the test purpose,
-	RRC connection resume: when RRCResume or RRCRelease message is part of the test purpose,
-	RRC connection re-establishment: when RRCReestablishment message is part of the test purpose
-	Special cases: e.g. when no contention resolution shall be sent according to the test purpose.



	Summary of change
	Added TC 8.1.6.1.3.1 to the list of TCs doing Early Contention

	TTCN module
	TestcaseProperties.ttcn

	MCC160 Comment
	



Before Change:
	function f_GetTestcaseAttrib_EarlyContentionResolution(charstring p_Testcase) return boolean
  {
    select (p_Testcase) {
      case ("TC_11_3_6_NR5GC") { return true; }
      case ("TC_6_4_1_1_NR5GC") { return true; }
      case ("TC_6_4_2_2_NR5GC") { return true; }
      case ("TC_7_1_1_1_3_NR5GC") { return true; }
      case ("TC_7_1_1_1_5_NR5GC") { return true; }
      case ("TC_8_1_1_1_1_NR5GC") { return true; }
      case ("TC_8_1_1_2_1_NR5GC") { return true; }
      case ("TC_8_1_1_2_3_NR5GC") { return true; }
      case ("TC_8_1_1_4_1_NR5GC") { return true; }
      case ("TC_8_1_1_4_2_NR5GC") { return true; }
      case ("TC_8_1_2_1_4_NR5GC") { return true; }
      case ("TC_8_1_3_1_23_NR5GC") { return true; }
      case ("TC_8_1_4_1_5_NR5GC") { return true; }
      case ("TC_8_1_4_1_6_NR5GC") { return true; }
      case ("TC_8_1_5_6_1_NR5GC") { return true; }
      case ("TC_8_1_5_8_1_NR5GC") { return true; }
      case ("TC_8_1_5_8_2_1_NR5GC") { return true; }
      case ("TC_8_1_5_8_2_2_NR5GC") { return true; }
      case ("TC_8_1_5_8_2_3_NR5GC") { return true; }
      case ("TC_8_1_6_1_3_3_NR5GC") { return true; }
      case ("TC_8_1_6_1_4_1_NR5GC") { return true; }
      case ("TC_8_1_6_1_4_2_NR5GC") { return true; }
      case ("TC_8_1_6_1_4_4_NR5GC") { return true; }
    }
    return false;
  }



After Change:
	function f_GetTestcaseAttrib_EarlyContentionResolution(charstring p_Testcase) return boolean
  {
    select (p_Testcase) {
      case ("TC_11_3_6_NR5GC") { return true; }
      case ("TC_6_4_1_1_NR5GC") { return true; }
      case ("TC_6_4_2_2_NR5GC") { return true; }
      case ("TC_7_1_1_1_3_NR5GC") { return true; }
      case ("TC_7_1_1_1_5_NR5GC") { return true; }
      case ("TC_8_1_1_1_1_NR5GC") { return true; }
      case ("TC_8_1_1_2_1_NR5GC") { return true; }
      case ("TC_8_1_1_2_3_NR5GC") { return true; }
      case ("TC_8_1_1_4_1_NR5GC") { return true; }
      case ("TC_8_1_1_4_2_NR5GC") { return true; }
      case ("TC_8_1_2_1_4_NR5GC") { return true; }
      case ("TC_8_1_3_1_23_NR5GC") { return true; }
      case ("TC_8_1_4_1_5_NR5GC") { return true; }
      case ("TC_8_1_4_1_6_NR5GC") { return true; }
      case ("TC_8_1_5_6_1_NR5GC") { return true; }
      case ("TC_8_1_5_8_1_NR5GC") { return true; }
      case ("TC_8_1_5_8_2_1_NR5GC") { return true; }
      case ("TC_8_1_5_8_2_2_NR5GC") { return true; }
      case ("TC_8_1_5_8_2_3_NR5GC") { return true; }
      case ("TC_8_1_6_1_3_1_NR5GC") { return true; }
      case ("TC_8_1_6_1_3_3_NR5GC") { return true; }
      case ("TC_8_1_6_1_4_1_NR5GC") { return true; }
      case ("TC_8_1_6_1_4_2_NR5GC") { return true; }
      case ("TC_8_1_6_1_4_4_NR5GC") { return true; }
    }
    return false;
  }

	





1. [bookmark: _Toc58491298][bookmark: _Toc60751532][bookmark: _Toc60757078][bookmark: _Toc85580846]Branches Executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc51622108][bookmark: _Toc56072723][bookmark: _Toc85580847]On Anritsu, the test case was executed on NR band n41 using NR ciphering nea2 and integrity algorithm nia2.
On Keysight, the test case was executed on NR band n41 using NR ciphering nea1 and integrity algorithm nia1.
1. Execution Log Files 
3.1 [bookmark: _Toc85580848]Qualcomm SM 8450 + SDX65
The Qualcomm SM 8450 + SDX65 UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Anritsu\ TC_8_1_6_1_3_1_LOG.xml
Keysight\TC_8_1_6_1_3_1_LOG.html
· PICS/PIXIT parameter file:
Anritsu\ TC_8_1_6_1_3_1_PIXIT.txt
Keysight\ TC_8_1_6_1_3_1_PIXIT.xml

4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc51622110][bookmark: _Toc56072725][bookmark: _Toc85580849]References
R5s211639 - Supporting information for addition of NR5GC R16 SON-MDT test case 8.1.6.1.3.1

	
	



